

VBA TEMELLERİ

1. Kodlar Sub blokları içerisinde yer almalıdır.
2. Excel tarafında hücre seçmek için

```
Range(Cells(1,1),Cell(6,6)).Select  
Veya  
Range("A1:F6").Select
```

3. Değişkenleri tanımlarken dikkatli olunmalıdır. Örneğin **integer** değişken -32000 ile +32000 arasında değer aldığından, bir işlemin sonucu bu değerden büyük olursa program **Over Flow** hatası verecektir.
4. Bir hücreye işlemin sonucunu/değer yazdırmak için

```
Dim as deger1 Integer  
Dim as deger2 Integer  
sonuc=deger1+deger2  
Range("A1") = sonuc  
Veya  
Cells(1,1) = sonuc
```

5. **OFFSET** kullanımı: Bu komut ile yazdırmak istediğimiz bir sonucu excelde istediğimiz hücreye konum belirterek taşıyabiliyoruz.

Ok işaretli -> satıra bakacak olursak yeni bir kod olarak ActiveCell'i görüyoruz.

Bunun anlamı, excelde o anda seçili olan hücreyi göstermektedir.

Offset(1,2)'nin anlamı ise; bu seçili hücreden 1 satır aşağı, 2 sütun sağa sonuç değerini yaz demiş oluyoruz.

```
Sub toplam()  
Dim sonuc As Long  
deger1 = 25000  
deger2 = 20000  
sonuc = deger1 + deger2  
Range("A2") = sonuc / 2  
->ActiveCell.Offset(1, 2) = sonuc  
End Sub
```

EXCEL NESNE MODELİ

Excel VBA nesne tabanlıdır ve excel içerisindeki her şey de bir nesnedir. Örneğin bir metin kutusu nesnedir, bu nesnenin özellikleri; uzunluğu, yüksekliği, rengi, yazı tipi, font indexi vs'dir.

1. Exceldeki nesne hiyerarşisi

```
Excel.Application.Workbooks("Kitap1").Worksheets("Sayfa1").Range("A1")
```

->Anlamı:

Excel->Uygulamasında->ÇalışmaKitapları içerisinde Kitap1'de->Sekmelerden Sayfa1'de->A1 hücresi

Kod yazarken Excel varsayılan olarak yukarıdaki;

```
Excel.Application.Workbooks("Kitap1").Work
```

Kısmını yazılı kabul eder. Dolayısı ile biz yazarken;

```
Worksheets("Sayfa1").Range("A1")
```

Şeklinde yazarız.

2. Bir sekmeden veya sayfadan başkaya bir sayfaya/sekmeye veri gönderme

```
Worksheets("Sayfa2").Range("A1")="veri buraya gönderildi"
```

Bu macro bir butona atanıp butona tıklandığında, sayfa2'deki A1 hücresine bu metin yazılır.

Burada önemli olan, Sayfa2 olarak isimlendirilen sekmenin adı kullanıcı vs tarafından değiştirilirse bu kodu yazdığımızda hata alır. Dolayısı ile şu şekilde yazarsak;

```
Worksheets(2).Range("A1")
```

->2. Sıradaki sekmenin A1 hücresine veri yazmış oluruz.

Eğer sekmenin sırası değiştirilirse de bu sefer yine hata alırız ya da yukarıdaki örnekteki gibi 2. Sıradaki sekme hangisi ise veri ona gönderilmiş olur.

Bu durumda da, VBA kod kısmında Properties'de Name kısmında sekme adı için ne yazıyorsa, bunu kodumuza yerleştirmemiz gerekir.

Sayfa2.Range("A1")="veri buraya gönderildi"

->Bu şekilde sekmenin ismi de, sırası da değiştirilse sorun olmayacaktır.

3. Koleksiyonlar çok kullanışlıdır. Nesnelere arasında aynı anda işlem yapabilmek ve bunları bir döngü içerisinde kullanabilmeye yarar. Koleksiyonlar; aynı tip özellikleri taşıyan birden fazla nesnenin bir araya gelmesi ile oluşur. Mesela; **sekmeler**.

*Nesneler arasında işlem yapabilmek için genellikle **For-Each** döngüsünü kullanırız.*


Örneğin; birden fazla şekil ekleyip (aynı boyut ve iç renginde) bunlardan istediğimizin mesela genişliğini değiştirmek istediğimizde aşağıdaki kodu kullanırız;

- a. Önce sayfa1 sekmesine bir buton ekliyoruz ve buna çift tıklayarak kod kısmını açıyoruz.

```
Private Sub CommandButton1_Click()
Dim sekil As Excel.Shape
For Each sekil in Shapes
If sekil.Name="Sekil5" Then (6-7 tane sekil oluşturup isim vermiştik excel sayfasında formül çubuğunun sol tarafındaki yerden)
Burada adı Sekil5 olan nesneyi değiştirmek istediğimizi belirtiyoruz
Sekil.Width=250
End If
Next
End Sub
```

WORKSHEET/WORKBOOK KOD BLOKLARI

PROSEDÜRLER



1. Yukarıdan da görüleceği üzere Sekmedeki (Sayfa1 vs) değil de çalışma kitabının kendisi ile ilgili yapacağımız kodla ilgili eylemlerde kullanıyoruz. Bu durumda Soldaki kısımdaki **BuÇalışmaKitabı**'nı seçiyoruz, yukarıda solda **Workbook**'u seçtikten sonra sağdaki eylemler karşımıza çıkar. Mesela, **BeforeClose** eylemini seçerek, çalışma kitabını kapatmadan önce bize bir mesaj göstermesini isteyebiliriz.

```
Private Sub Workbook_BeforeClose(Cancel As Boolean)
MsgBox "Kapatmak istediğinize emin misiniz?"
End Sub
```

2. Bu durumun aynısı elbette Sayfa1 yani sekmelerde iken de geçerli ve buna göre eylemler çıkmaktadır.

```
Private Sub Worksheet_SelectionChange(ByVal Target As Range)
```

```
End Sub
```

3. Ödev1 bu bölümde yapıldı.

MANTIKSAL SINAMALAR

1. IF, THEN, ELSE Yapısı

Örneğin yapılan satış miktarına göre prim hesaplaması

```
Sub If_Ornegi()
If Range("B1")>1 and Range("B1")<=100 Then
Range("C1")=Range("B1")*1.05

Elseif Range("B1")>100 and Range("B1")<=500 Then
Range("C1")=Range("B1")*1.08

Elseif Range("B1")>500 and Range("B1")<=1000 Then
Range("C1")=Range("B1")*1.10

Else Range("C1")=Range("B1")*1.15
Endif
End Sub
```

2. Ödev2 Burada...A1:C5 hücreleri arasında bir yerde herhangi bir değişiklik yapmaya çalıştığımızda mesaj kutusu uyarısı görüntülensin.

*Burada dikkat etmemiz gereken nokta aktif sekmemiz hangisi ise onun içerisindeki kod bloğuna yazmamız gerektiği. Ayrıca **KOD BLOKLARI** bölümündeki resimde görüldüğü gibi, eylemlerden de **Change'in** seçili olmasıdır. A1 ve C5 arasındaki hücreleri satır ve sütun indirleri ile belirliyoruz (Odan başlıyorlar diğer yazılımlardaki gibi ve A, C vs gibi isimlendirmeler kullanmıyoruz bu durumda, sadece alanımızı içine alan **SATIR ve SÜTUN İNDİSLERİNİ** belirleyerek giriyoruz.)*

```
Private Sub Worksheet_Change(ByVal Target As Range)
If Target.Row < 6 And Target.Column < 4 Then
MsgBox "Değişiklik yapmaya çalışıyorsunuz!"
End If
End Sub
```

3. Ödev3 Burada...Verilen tabloda hangi hücreye tıklanırsa, o hücrenin ait olduğu satır tablo boyunca farklı bir renk olsun.

Burada yukarıdaki ömekten farklı olarak **SelectionChange** kullanılmıştır. Yani bir hücreden bir diğerine tıklayarak geçtiğimiz andaki değişikliği belirtmiş oluyoruz.

```
Private Sub Worksheet_SelectionChange(ByVal Target As Range)
```

```
'Tablodaki A2:D7 arasını hedef olarak gösteriyoruz.
```

```
If Target.Row < 8 And Target.Row > 1 And Target.Column = 1 Then
```

```
Range("A1") = Target.Value
```

```
End If
```

```
End Sub
```

Yukarıdaki kod ile aşağıdaki resimde olduğu gibi; A1 hücresinde, tabloda hangi hücreye tıklanıyorsa o değer görünmektedir.

	A	B	C	D
1	Onstagram			
2	Firma İsmi	Yetkili Kişi	Vergi No	Matrah
3	Mwitter	Andrea Müller	2536489653	5,246,523,156,326
4	Casebook	Adam Smith	5265448963	2,353,121,365,313
5	Onstagram	Hale Çay	6635326656	1,212,142,142,313
6	Meytube	Furkan Kasa	3659653265	7,412,124,721,231
7	Taktik	Lale Pay	8465545154	5,454,231,654,321

Ama burada `Target.Column=1` ifadesine dikkat etmeliyiz. Sadece A sütunundaki değerlere ve tablo alanındaki değerlere tıkladığımızda değer görünecektir.

Peki tablodaki seçili hücrenin tüm satırını nasıl renklendireceğiz?

Burada da tablo alanını seçerek, koşullu biçimlendirme yapıyoruz ve ilgili satır sarı renge boyanmış olacaktır. Aşağıdaki gibi.

	A	B	C	D	E	F	G	H	I	J	K	L
1	Casebook											
2	Firma İsmi	Yetkili Kişi	Vergi No	Matrah								
3	Mwitter	Andrea Müller	2536489653	5,246,523,156,326								
4	Casebook	Adam Smith	5265448963	2,353,121,365,313								
5	Onstagram	Hale Çay	6635326656	1,212,142,142,313								
6	Meytube	Furkan Kasa	3659653265	7,412,124,721,231								
7	Taktik	Lale Pay	8465545154	5,454,231,654,321								
8												
9												
10												
11												
12												
13												
14												
15												
16												
17												
18												
19												

Biçimlendirme Kuralını Düzenle

Kural Türü Seçin:

- ▶ Değerlerine bağlı olarak tüm hücreleri biçimlendir
- ▶ Yalnızca şunu içeren hücreleri biçimlendir
- ▶ Yalnızca derecelendirilen en üst veya en alt değerleri biçimlendir
- ▶ Yalnızca ortalamının üstünde veya altında olan değerleri biçimlendir
- ▶ Yalnızca benzersiz veya yinelenen değerleri biçimlendir
- ▶ Biçimlendirilecek hücreleri belirlemek için formül kullan

Kural Açıklamasını Düzenleyin:

Bu formül doğru olduğunda değerleri biçimlendir:

=SAŞ1=SAŞ3

Önizleme: AaÇçĞğŞşZz

Tamam iptal

4. **İç İç IF Yapısı Örneği...** Burada Ürün Elektronikse, KDV'sini %5'ten hesaplayacak ve eğer fiyat+KDV 1000 TL'nin altında ise Taksit sayısına 9 veya 12 diyecek.

	A	B	C	D	E
1	Ürün Kategorisi	Fiyat	KDV	Tutar	Taksit
2	Elektronik	100.00	5%	105.00	12

Private Sub Worksheet_SelectionChange(ByVal Target As Range)

If Range("A2") = "Elektronik" Then

Range("C2") = "%5"

Else

Range("C2") = "%8"

End If

Dim tutar As Integer

tutar = Range("B2") + Range("B2") * Range("C2")

Range("D2") = tutar

If tutar < 1000 Then

Range("E2") = 12

Else

Range("E2") = 9

End If

End Sub

5. **Select Case Yapısı Örneği...** D2 hücresine girilen not karşılığına gelen Harf değerini yazdırmak.

	A	B	C	D	E	F	G	H	I
1	SKALA			Puan	Puan Karşılığı Harf				
2	Puan	Harf		72	BB				
3	0-30	FF							
4	31-50	DD							
5	51-70	CC							
6	71-90	BB							
7	91-100	AA							

Harf Karşılığını Bul

```
Private Sub CommandButton1_Click()

    puan = Range("D2")
    Select Case puan
    Case 0 To 30
        Range("E2") = "FF"
    Case 31 To 50
        Range("E2") = "DD"
    Case 51 To 70
        Range("E2") = "CC"
    Case 71 To 90
        Range("E2") = "BB"
    Case 91 To 100
        Range("E2") = "AA"
    Case Else 'Yukarıdaki Hiçbir Durumun Olmaması durumunda'
        Range("E2") = "Hatalı Not"
    End Select
End Sub
```

MSGBOX/ BOX TÜRLERİ

1. Mesaj Box'lar ile ekrana uyarılar verebiliriz.

2. Mesaj boxlardaki uyarı sonuçlarına göre işlem yaptırabiliriz.
3. Temel yaygın kullanılan özellikler aşağıdaki kodlardaki gibidir.

```
Sub mesaj_kutulari()

MsgBox "İşlemi Yapmak istiyor musunuz?", vbCritical, "Uyarı"
MsgBox "Bu işlemi yapamazsınız", vbInformation, "Uyarı"
MsgBox "Bu işlemi yapamazsınız", vbAbortRetryIgnore, "Uyarı"
MsgBox "Bu işlemi yapamazsınız", vbYesNo, "Uyarı"


'değişkene atarken parantez koyuyoruz dikkat!
cevap=MsgBox ("Bu işlemi yapmak istiyor musunuz?", vbYesNo+vbCritical, "Uyarı")
    if cevap=vbYes Then
        MsgBox "İşlem Yapılmadı"
    Else
        MsgBox "İşlem Yapıldı"
    End If

'Veri Girişi Yaptırma ve sayının karesini alma
girdi=MsgBox "Bir değer girin", vbInformation, "Veri Girişi"
MsgBox girdi*girdi


End Sub
```

4. Birden fazla satırda mesaj göstermek için;

```
a = Date
b = Time
c = Name
MsgBox a & vbNewLine & b & vbNewLine & c
```



Veya aşağıdaki gibi



```
Private Sub Workbook_Open()  
iResult = MsgBox("This information is on the first line. " & vbCrLf & "This in  
formation is on the 2nd line. " & vbCrLf & _  
    "Do you wish to continue?", vbYesNo + vbInformation, "Message Box")  
End Sub
```

5. **Ödev4.** Kullanıcı Excel'i açtığında, kullanıcı adı, bugünün tarihi ve bugünün saati alt alta olacak şekilde ekrana bir Hoş geldin penceresi açtırmak.

```
Private Sub Workbook_Open()  
'MsgBox "Hoşgeldiniz, ", vbInformation, "Hoşgeldin"  
a = Date  
b = Time  
c = Application.UserName  
  
MsgBox a & vbCrLf & b & vbCrLf & c  
End Sub
```

DEBUG PRINT

1. Yazılımcının kodlarının çalışabilirliğini kontrol etmesi için **Debug** kullanılır.
2. Kod inceleme yönetimi...
3. MessageBox'ları her zaman kullanmak yerine bunu kullanmak daha etkilidir. MsgBox'lar programı yavaşlatabilmektedir.

(General) debug_print

```
Sub debug_print()
a = 5
b = 6
c = a * b
Debug.Print c
End Sub
```

Debug Print kodu ile de Immediate penceresinde kodumuzun sonucunu görürüz (excel ekranına gitmeye gerek kalmadan)

F5 yapılırsa kodu adım adım Locals penceresinde görürüz.

Immediate

```
30
30
```

Locals

Expression	Value
Me	
a	5
b	6
c	30

DÖNGÜLER

1. For-Next Döngüsü

```
Private Sub CommandButton1_Click()
Dim i As Integer

For i = 1 To 10
Cells(i, 1) = i * 10
Next i

End Sub
```

A sütununda ilk 10 hücre satırına değer yazdırmış oluyoruz i ile indis değeri oluşturarak. Eğer 1er 1er artmasını değil de istediğimiz miktarda artmasını istersek de aşağıdaki kodu kullanacağız.

```
Private Sub CommandButton1_Click()
Dim i As Integer
```

```
For i = 1 To 10 Step 2
Cells(i, 1) = i * 10
Next i

End Sub
```

Arttırarak değil de geriye doğru sayarak yapmak istersek

```
Private Sub CommandButton1_Click()
Dim i As Integer

For i = 10 To 1 Step -1
Range("A" & i) = i * 10
Next i

End Sub
```

2. A1:A20 hücrelerinde bulunan 1000-5000 arası rastgele sayıların Tek mi çift mi olduğunu bulan ve B2:B20 hücrelerinde karşılıklarına yazan kod aşağıdaki gibidir.

```
Private Sub CommandButton1_Click()
Dim i As Integer

For i = 1 To 20
If Cells(i, 1) Mod 2 = 0 Then
Cells(i, 2) = "Çift"
Else
Cells(i, 2) = "Tek"
End If
Next i

End Sub
```

*Burada Cells(i,1)-> hep A sütununda ilerlerken,
Cells(i,2)-> hep B sütununda ilerlemektedir.*

3. **ÖDEV5...** yukarıdaki tek çift örneğini bu sefer bizim verdiğimiz from 1 to 20'ye değil de. Hücre eklendikçe kendi hesaplayacak şekilde yapacağız. **SON DİNAMİK SATIRI BULMA**

```
Private Sub CommandButton1_Click()
Dim i As Integer
```

```
'Aşağıdaki her iki komut da sütundaki son hücreye kadar döngüyü götürmektedir.  
For i = 1 To Range("A" & Rows.Count).End(xlUp).Row  
'For i = 1 To Cells.Find("*", , , , xlByRows, xlPrevious).Row  
    If Cells(i, 1) Mod 2 = 0 Then  
        Cells(i, 2) = "Çift"  
    Else  
        Cells(i, 2) = "Tek"  
    End If  
Next i  
End Sub
```

Veya daha kısa bir yol olarak aşağıdaki kodu kullanabiliriz.

```
Private Sub CommandButton2_Click()  
    Dim i As Integer  
  
    'veya daha kısa yol olarak  
    sonsatir = Cells(Rows.Count, "A").End(xlUp).Row  
    For i = 1 To sonsatir  
        If Cells(i, 1) Mod 2 = 0 Then  
            Cells(i, 2) = "Çift"  
        Else  
            Cells(i, 2) = "Tek"  
        End If  
    Next i  
End Sub
```

4. ÖDEV 6: öğrencilerin aldıkları notlara karşılık gelen harf değerlerini notların karşısına yazma. Listeye isim eklendikçe hesaplama yapması.

	A	B	C	D	E	F	G
1	İsim	Puan	Harf			SKALA	
2	ALİ	96	AA			Puan	HARF
3	MAHMUT	12	FF			0-30	FF
4	MANSUR KÜRŞAD	47	DD			31-50	DD
5	GAMZE	78	BB			51-70	CC
6	MİRAÇ	77	BB			71-90	BB
7	YÜCEL	53	CC			91-100	AA
8	KUBİLAY	96	AA				
9	HAYATİ	82	BB				
10	BEDRİYE MÜGE	38	DD			CommandBut	
11	BİRSEN	67	CC				
12	SERDAL	31	DD				
13	BÜNYAMİN	54	CC				
14	ÖZGÜR	43	DD				
15	FERDİ	11	FF				
16	BEVUAN	15	CC				

```

Private Sub CommandButton1_Click()
Dim i As Integer
sonsatir = Cells(Rows.Count, "B").End(xlUp).Row

For i = 2 To sonsatir
    puan = Cells(i, 2)
    Select Case puan
    Case 0 To 30
        Cells(i, 3) = "FF"
    Case 31 To 50
        Cells(i, 3) = "DD"
    Case 51 To 70
        Cells(i, 3) = "CC"
    Case 71 To 90
        Cells(i, 3) = "BB"
    Case 91 To 100
        Cells(i, 3) = "AA"
    Case Else
        Cells(i, 3) = "Hatalı not"
    End Select
Next i
End Sub

```

5. ÖDEV 7: Ürün barkodlarında numarası B ile başlayanların arka planının sarı, C ile başlayanların ise kırmızı olmasını istiyoruz. Hücredeki kelimenin, değerin ilk karakterini harfini bulma ile yapacağız.

	A	B	C	D
1	Ürün Kodu			
2	A23984092			
3	B28342983		1. Yöntem	
4	A93480349			
5	C79874329		2. Yöntem	
6	B20398402			
7	A98420239			
8	C92384023			
9	B20398409			

1. Yöntem

```
Private Sub CommandButton1_Click()
    sonsatir = Cells(Rows.Count, "A").End(xlUp).Row
    'veya aşağıdaki gibi yazılabilir
    'sonsatir = Cells(Rows.Count, 1).End(xlUp).Row

    For i = 2 To sonsatir
        If VBA.Left(Cells(i, 1).Value, 1) = "B" Then
            Cells(i, 1).Interior.Color = 65535
        ElseIf VBA.Left(Cells(i, 1).Value, 1) = "C" Then
            Cells(i, 1).Interior.Color = 255
        End If
    Next i
End Sub
```


2nci Yöntem Kısa Kod: Burada "B*" anlamı: B ile başlayan ama gerisi önemli olmayan demek.

"*B" yazsaydık B ile biten ama ilk karakterleri önemli olmayan demek.

```
Private Sub CommandButton2_Click()
    sonsatir = Cells(Rows.Count, "A").End(xlUp).Row
    'veya aşağıdaki gibi yazılabilir
    'sonsatir = Cells(Rows.Count, 1).End(xlUp).Row

    For i = 2 To sonsatir
        If Cells(i, 1) Like "B*" Then
            Cells(i, 1).Interior.Color = 65535
        ElseIf Cells(i, 1) Like "C*" Then
            Cells(i, 1).Interior.Color = 255
        End If
    Next i
End Sub
```

6. ÖDEV8: Input box ile kullanıcıdan girilecek değere göre 5x5 veya herhangi girilen bir değerde sıralı matris oluşturulacak.



```
Private Sub CommandButton1_Click()
Dim a As Integer
Dim b As Integer
Dim k As Integer

a = InputBox("Satır Sayısı Girin")
b = InputBox("Sütun Sayısı Girin")
k = 1
For i = 1 To a
    For j = 1 To b
        Cells(i, j) = k
        k = k + 1
    Next j
Next i
'Sayfa içeriğini temizlemek için
c = MsgBox("matrisi temizlemek ister misiniz?", vbYesNo)
If c = vbYes Then
ActiveSheet.UsedRange.ClearContents
Else
End If
End Sub

'Sayfa içeriğini butonla temizlemek için
Private Sub CommandButton2_Click()
c = MsgBox("matrisi temizlemek ister misiniz?", vbYesNo)
If c = vbYes Then
ActiveSheet.UsedRange.ClearContents
Else
End If
End Sub
```

7. Do Loop, Loop While, Loop Until ve While Wend Döngüsü

Örneğin ekranda renk kodlarını sıra ile hücelere yazdırıp, hüceleri renklendiriyoruz.

Bu hüceler belirli süre aralıklarla sırayla ekranda belirsin istiyoruz.

	A	B	C	D	E	F	G	H	I	J
1		2	3							
2	4	5	6				Renk İndeksi Oluştur			
3	7	8	9							
4	10	11	12				Sayfayı Temizle			
5	13	14	15							
6	16	17	18							
7	19	20	21				Satir ve Sütun Sayisi			
8	22	23	24							
9	25	26	27							
10	28	29	30				Renk İndeksi Timer Ile			
11	31	32	33							
12	34	35	36							
13	37	38	39				Renk İndeksi Timer Ile Alternatif			
14	40	41	42							
15	43	44	45							

```


Private Sub CommandButton5_Click()
'Bu yöntemde hücreler sırası ile belirlenen zaman aralığında
'(1/3 saniye kadar) bekledikten sonra renklenmektedir. Optimizasyon
'çalışmasında blokların sıralı üretimi noktasında kullanabiliriz. Ancak
'bu yöntemde de ESC ile koddan çıkamadım.
For i = 1 To 56
  Start = Timer
  Do
  DoEvents
  Finish = Timer
  With Cells(Int((i - 1) / 3) + 1, (i - 1) Mod 3 + 1)
    .Interior.ColorIndex = i
    .Value = i
    .HorizontalAlignment = xlCenter
    .Borders.Weight = xlThin
    .Borders.LineStyle = xlContinuous
  End With
  Loop While Finish - Start < (1 / 3)
Next i
End Sub

```


8. For Each Döngüsü

Nesneler arasında döngü kurmak için kullanıyoruz.

Örneğin excel sayfasına 5 tane dörtgen çizelim ve isimlerini sol üst köşeden değiştirelim.



Bu şekillerin renklerini kod ile değiştirecek olursak.



```
Private Sub CommandButton1_Click()
'şekiller üzerinde işlem yaptığımız için değişkenimizi şekil olarak
'tanımlıyoruz
'Sekmeler yani worksheetler üzerinde işlem yapacaksak değişkenimizi
'Worksheet olarak tanımlamamız gerekiyor
Dim shp As Shape
Dim ws As Worksheet

For Each shp In Shapes
shp.Fill.ForeColor.RGB = RGB(249, 249, 15)
Next
End Sub
```

*Burada görüleceği üzere nesnelere üzerinde işlem yaptığımız için **değişkenlerimizi** de buna göre tanımlıyoruz.*

Diğer bir örnek olarak; Sayfa2'de sadece değer barındıran hücreleri 2'ye bölerek aynı hücrelere yazdırmak istersek eğer

	A	B	C	D	E	F	G	H
1			4.15625					
2			2.421875					
3			12.48438					
4			12.96875					
5			5.609375					
6			8.46875					
7			4.765625					
8			6.09375					
9			4.53125					

Deger İçeren Hücreleri 2ye Böl

```
Private Sub CommandButton1_Click()
Dim ws As Worksheet
Dim rng As Range

'ws'nin Sayfa2 olduğunu burada ayarlıyoruz
Set ws = Sheets(2)

'UsedRange komutu ile de, Sayfa2'de içinde değer barındıran
'hücreler içerisinde işlem yapmasını sağlıyoruz.
For Each rng In ws.UsedRange
rng.Value = rng.Value / 2
Next

End Sub
```

9. Ödev9: 5 adet sekme açıp 1'den 5'e yazı ile isimlendir ve 1. Sekme hariç diğer sekmeleri bir butonla gizle ve başka bir butonla da görünür hale getir.

```
Private Sub CommandButton1_Click()
Dim ws As Worksheet

For Each ws In Worksheets
'Sekmenin adı Sayfa1 değilse gizle
If ws.Name <> "Sayfa1" Then
ws.Visible = xlSheetHidden
End If
Next

End Sub

Private Sub CommandButton2_Click()
Dim ws As Worksheet

For Each ws In Worksheets
ws.Visible = xlSheetVisible
Next

End Sub
```

```
End Sub
```

10. On Time İle Döngü

Bu döngü komutu ile başka bir prosedüre gönderme yapabiliriz.

```
Sub hatirlatma()
MsgBox "vakti geldi"
End Sub

Sub zamanlama()
Application.OnTime TimeValue("13:51:00"), "hatirlatma"
End Sub
```

Yukarıda görüleceği üzere hatirlatma isimli prosedürümüzü aşağıda zamanlama içerisinde OnTime ile belirlediğimiz zamanda çalışması üzere ayarlamış olduk. Buna göre VBA'da yeni bir modül açtık, excelde bir buton oluşturup yukarıda kodu yapıştırdığımız prosedürü bu butona makro atayarak çalışmasını sağladık. Bu şekilde butona tıkladığımızda ve TimeValue değerinde belirlediğimiz zaman geldiğimizde ekrana "vakti geldi" şeklinde mesaj penceresi açılmış olacaktır.

Bu şekilde başka kodlar veya eylemleri de zamanlayabiliriz.

Mesela excel hücrelerinde gerçek zamanlı çalışan bir saat bu şekilde yerleştirebiliriz.



Butona makro ata özelliğini yapamadım o yüzden kodları modül içerisine yazarak, daha sonra şekle sağ tık makro ata diyerek kodu çalıştırabiliyoruz.

```
Sub saat()
Application.OnTime Now + TimeValue("00:00:01"), "saat"
Range("A1") = Time
End Sub
```

ActiveX'ten buton eklediğim için olmuyormuş. Forms'dan eklersem macro ata oluyor.
ENTERESAN!!!

11. Goto ve On Error Resume Next

Hata içeren hücre yapısında kodun takılmaması ve hata vermemesi için bu kısımların atlanmasını sağlar.

	A	B	C	D	E
1	32.40625				
2	52.92188				
3	56				
4	35.63281			CommandBut	
5	26.77344				
6	31.21875				
7	57.1875				
8	a4545				
9	6108				
10	7731				

```
Private Sub CommandButton1_Click()
On Error GoTo kurtul
For i = 1 To 18
Cells(i, 1) = Cells(i, 1) / 2
Next i
kurtul:
End Sub
```

Yukarıdaki koda göre eğer **On Error Goto** kodunu kullanmamış olsaydık, kodumuz çalışırken A8 hücresine geldiğinde hata verecektir. Bu şekilde A8 hücresinde hatayı gördüğümüz an **kurtul:** olarak belirttiğimiz kısa gitmektedir.

Ancak bu durumda da kalan hücrelerde hesaplama yapmamaktadır.

Bunu engellemek için de yeniden bir buton ve aşağıdaki kod ile bunun üstesinden geliyoruz.

	A	B	C	D	E
1	0.031647				
2	0.051682				
3	0.054688				
4	0.034798			on goto	
5	0.026146				
6	0.030487			resume next	
7	0.055847				
8	a4545				
9	47.71875				
10	60.39844				

```
Private Sub CommandButton2_Click()
On Error Resume Next
For i = 1 To 18
Cells(i, 1) = Cells(i, 1) / 2
Next i
End Sub
```

12. Komuttan çıkmak için **Exit** komutunu kullanabiliriz.

	A	B	C	D	E	F	G	H	I	J	K
1	0.031647						3273	18520			
2	0.051682						1653	13365			
3	0.054688						3096	13230			
4	0.034798			on goto			1795	7665			
5	0.026146						a4545				
6	0.030487			resume next			1796				
7	0.055847						2376				
8	a4545						4932				
9	47.71875			exit olayı							
10	60.39844										
11	a4555										

Microsoft Excel

5 satırınsa metinsel ifade bulunmaktadır

Tamam


```
Private Sub CommandButton3_Click()
For i = 1 To 8
If IsNumeric(Cells(i, 7)) Then 'IsNumeric komutu ile hücrenin sayısalılığı kontrol ediliyor
Cells(i, 8) = Cells(i, 7) * 5
Else
MsgBox i & " satırında metinsel ifade bulunmaktadır"
Exit For 'buraya dikkat işte burada for döngüsünden çıkıyoruz!
End If
Next i
End Sub
```

Yukarıdaki kodda görüleceği üzere hücredeki değerın nümerik olup olmadığını **IsNumeric** kodu ile kontrol ediyoruz. Nümerikse işlemi yapıyor değilse ekrana uyarı veriyor ve for döngüsünden çıkıyor.


ÖNEMLİ: Burada Exit komutu için Exit for yerine Exit Sub da yazabilirdik. Veya mesela Do döngüsü kullanmışsak Exit Do da yazabilirdik.


KULLANICI TANIMLI FONKSİYONLAR

1. **ÖNEMLİ:** Fonksiyonlar MODULE içerisine yazılır.



2. **ÖNEMLİ:** Fonksiyonları tüm excellerde çalışması için eklenti olarak kaydedebiliriz.






Aşağıdaki dosya yolu içerisine kaydedildiğine dikkat etmeliyiz.

<C:\Users\furka\AppData\Roaming\Microsoft\AddIns>

Daha sonra bunu eklentileri etkinleştirerek etkin hale getirip, excellerimizde kullanabiliyoruz.



13. Ödev10: A1 ve A2 hücresine girilen değerler arasındaki çift sayıların toplamını A4 hücresine fonksiyonumuzun formülünü yazdığımızda sonucu verecek fonksiyon oluşturulacak.

	A	B	C	D	E	F
1	200					
2	5000					
3						
4	6242600					
5						

Fonksiyonumuza **cift_toplam** adını verdik.

```
Function cift_toplam(sayi1 As Range, sayi2 As Range)
cift_toplam = 0
For i = sayi1 To sayi2
  If i Mod 2 = 0 Then
    cift_toplam = cift_toplam + i
  End If
Next i
End Function
```

Hücelere formül veya değer yazma şeklinde yaptığımız için **sayi1** ve **sayi2**'yi **Range** olarak tanımladığımıza dikkat etmeliyiz.

- 14. Ödev11:** G,H,I hücrelerindeki renk değerlerini seçmiş olduğumuz aralıktaki hücrelerde arayarak, aynı renkteki hücrelerin içindeki sayısal değerleri renklerine göre toplayacak bir fonksiyon.

	A	B	C	D	E	F	G	H	I
1	4287	2474	2876	1375	2248				
2	2187	1621	1368	1461	3568				
3	4805	1732	4657	3614	2909				
4	4489	1975	3010	4200	1510		69844	82284	69470
5	2326	2285	2575	2063	2382				
6	2980	2285	1082	1183	3533				
7	4263	2285	1578	1155	4516				
8	4193	3692	3450	2564	1090				
9	3759	4634	4244	4154	3315				
10	2691	3372	3117	2737	3477				
11	3310	4087	3811	3482	4176				
12	3132	1031	3783	1067	3469				
13	4597	1720	3891	1854	1717				
14	3975	3050	2830	2544	4911				
15	4751	2578	4400	1622	2464				

```

Function renk_topla(aralik As Range, renk As Range)
'önce değer barındıran hücreleri seçeceğimiz için aralik isimli değişkeni
'Range olarak tanımladık. Sonrasında da hangi rengi seçeceksek onun bulunduğu
'hücreyi seçeceğimiz için renk isimli değişkeni de Range olarak belirledik.

'for döngüsü için kaç hücremiz varsa bunun adedini belirliyoruz
saydir = aralik.Count
'ilk değerimiz 0 elbette
renk_topla = 0

For i = 1 To saydir
'i.nci hücrenin iç renk indeksi, renk olarak seçilen hücrenin
'renk hücrelerine eşitse
If aralik(i).Interior.ColorIndex = renk.Interior.ColorIndex Then
'aralik.i'nci hücrenin değeri ile başlangıç değerini topla
renk_topla = renk_topla + aralik(i).Value
End If
Next i

End Function

```

VBA YERLEŞİK FONKSİYONLAR

1. Date, Time, Now, Format, CInt, Cstr, Cdate, Len, Strconv, Strreverse

```

Range("A1")= VBA.Date
Range("A1")= VBA.Time
Range("A1")= VBA.Now
Range("A1")= VBA.Format(Range("A1"),"#,##0.00")
'CInt özellikle useformlardaki textbox içerisindeki str değişkenini integer'a
dönüştürmek için kullanılır
'Cstr de aynı şekilde Integer veya nümerikten String formata
'Cdate tarihe
'Len hücre veya değişkendeği uzunluğu verir. Mesela hücrede Ali yazıyorsa 3 ol
arak döndürür
'Strreverse hücredeki yazılı kelimeyi tersten yazar.

```

2. ISDATE, ISNUMERIC, ISTEXT, ISEMPY

Bunlar da aslında birer koşuldur. VBA tarafından kolaylık olması bakımından tanımlanmıştır.


ISDATE: mesela hücredeki değer bir tarih mi? Değilse şunu yap gibi bir koşulu yazmamızı sağlar.

ISNUMERIC: hücredeki değer nümerik olup olmadığı koşulunu yazmamızı sağlar

ISTEXT: hücredeki değer metin mi?

IS EMPTY: hücre boş mu?

	A	B	C	D	E	F	G	H	I
1	ISDATE	1.01.2021		Isdate 1	Isdate 2				
2	ISNUMERIC			Isnumeric 1					
3	ISTEXT								
4	ISEMPTY								
5									
6					1.01.2021				



```

Private Sub CommandButton1_Click()
'***1nci Yöntem***
If IsDate(Range("B1")) Then
    MsgBox "giriş değeri tarihtir"
Else
    MsgBox "giriş değeri tarih değildir"
End If
End Sub

Private Sub CommandButton2_Click()
'***2nci Yöntem***
'2nci yöntem olarak, koşullarda sürekli Range("A1") vs şeklinde tekrarlamamak
adına alan olarak değer girilmiş olan hücremizi Range olarak tanımlayıp
sonra Set ile hangi hücremizi atadığımızı belirtiyoruz. Görüleceği üzere
alan1'de hücre Belirtirken, alan2'de aktif hücre hangisi ise onu belirtiyoruz.
Daha sonra Ifkoşulu içerisinde de bu değişkenlerimizin adı ile atıf yapıyoruz
sadece.
Dim alan1 As Range
Dim alan2 As Range
'Dim hucre_adresi As String

Set alan1 = Range("B1")
Set alan2 = ActiveCell
'hangi hücrenin aktif olduğunu değişkenimize atıyoruz
hucre_adresi = ActiveCell.Address

If IsDate(alan1) Then
    MsgBox "B1 hücresine girilen değer tarihtir"
Else
    MsgBox "B1 hücresine girilen değer tarih değildir"
End If

'aktif hücrenin ismi ile birlikte uyarısını yazdırıyoruz
If IsDate(alan2) Then
    MsgBox hucre_adresi & " hücresine girilen değer tarihtir"
Else
    MsgBox hucre_adresi & " hücresine girilen değer tarih değildir"
End If

```

Burada değinilen iki önemli şey **Dim ... as Range ve Set** komutları bunları unutmamak gerek.

3. VLOOKUP, SUMIFS, COUNTIFS (DüşeyAra, ÇokeTopla, ÇokEğerSay)

```
Sub vlookup()
Range("o3") = Application.WorksheetFunction.vlookup(Range("m3"), Range("A1:K26"), 3, 0)
End Sub

Sub sumifs()
Range("o7") = Application.WorksheetFunction.sumifs(Range("C:C"), Range("B:B"), Range("m7"))
End Sub

Sub countifs()
Range("o11") = Application.WorksheetFunction.countifs(Range("B:C"), Range("m11"))
End Sub
```

1. VLOOKUP: ("Aranan Değer", "Aranan Tablo Aralığı", "Aranan Sütun İndisi", "Tam eşleşme" -> Aranan değere karşılık gelen değeri verir.
2. SUMIFS: ("Toplanacak Aralık", "Aranacak Aralık", "Aranan Değer") -> Aranan değere sahip olanların toplamını verir.
3. COUNTIFS: ("Sayılacak Aralık", "Aranan Değer") -> Kaç adet olduğunu verir.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	
1	İrt No	Maden	Toplam (Ton)	Kaynak (Ton)	GORÜNÜR	Cr2O3	(%Pb)	(%Zn)	(g/t Ag)	(%Cu)	(g/t Au)		Ruhsat Ton	DüşeyAra/VLookUp			
2	İR.4225	Cr	340,185	250,000	90,185	17.81							Aranan	Excel	VBA		
3	İR.5259	Cr	46,555,552	17,149,000	29,406,552	6.03							İR.5259	46,555,552.00	46,555,552.00	VLookUp	
4	İR.5189	Cr	446,325	250,000	196,325	12.01											
5	İR.7060	Cr	48,000	37,000	11,000	14.91							Kaç ton	ÇokEğerTopla/Sumifs			
6	İR.1545	Cr	1,024,167	454,856	569,311	15.44							Aranan	Excel	VBA		
7	İR.7640	Cr	410,303	240,000	170,303	15.32							Cr	50,530,505.00	50,530,505.00	Sumifs	
8	İR.2016	Cr	10,000	10,000	0												
9	İR.85684	Cr	606,324	207,000	399,324	16.00							Kaç Ruhsat	ÇokEğerSay/Countifs			
10	İR.85681	Cr	734,565	336,000	398,565	24.77							Aranan	Excel	VBA		
11	İR.200807267	Cr	275,623	178,000	97,623	18.00							Cr		15	15.00	Countifs
12	İR.51955	Cr	10,500	10,000	500	48.00											

Burada VBA ekranında prosedürleri yazıp (sub sumifs() vs şeklinde olan), excelde şekil ekleyip sağ tık makro ata diyerek buton yerine bu yolu kullandık.

DİZİLER

1. Dizilerin indis sayısı her zaman 0'dan başlar.
2. Ama en üste eğer "Option Base 1" yazarsak, dizinin ilk elemanının indisi 1 olacaktır.
3. Bir diziyi aşağıdaki gibi oluşturabiliriz. Bu tek boyutlu bir dizidir.

```
Sub diziler1()
```

```
dizi1 = Array("elma", "armut", "kel", "tankut")
End Sub
```

4. Diziler aşağıdaki şekillerde tanımlanabilirler

```
Sub diziler2()

'Diziyi aşağıdaki gibi boş tanımlayabiliriz. Variant olarak tanımlanması ise
'e içerisinde
'nümerik veya text elemanlar olabileceği düşünülebilir.
dizi2() As Variant
'10 elemanlı tek boyutlu dizi
dizi3(10) As Variant
'10 elemanlı tek boyutlu
dizi4(1 To 10) As Variant
'2 satır 5 sütunlu 2 boyutlu integer dizi
dizi5(2, 5) As Integer
'3 boyutlu integer dizi
dizi6(2, 5, 3) As Integer

End Sub
```

5. Dizimize excelden eleman eklemek için aşağıdaki kodu kullanabiliriz.

```
Sub diziler3()
Dim dizi(1 To 15) As Integer
Dim i As Integer

'Burada Excel hücrelerinde yer alan değerlerimizi
'dizi isimli dizimizin içerisine tanımlıyoruz.
For i = 1 To 15
dizi(i) = Cells(i, 1)
Next i

'sonra da dizinin 11nci elemanını ekranda gösteriyoruz.
MsgBox dizi(11)

End Sub
```

	A	B
1	91	
2	11	
3	23	
4	27	
5	58	
6	35	
7	96	
8	86	
9	45	
10	66	
11	54	
12	95	
13	59	
14	26	
15	81	

6. Dizinin ilk ve son elemanı aşağıdaki kodlar ile ekrana yansıtılabilir.

```

Sub diziler4()
Dim dizi(1 To 15) As Integer
Dim i As Integer

'Burada Excel hücrelerinde yer alan değerlerimizi
'dizi isimli dizimizin içerisine tanımlıyoruz.
For i = 1 To 15
dizi(i) = Cells(i, 1)
Next i

'dizinin ilk elemanı (yani indisi) ve son elemanı (son satırı gibi)
'aşağıdaki kodlar ile bulunabilir.
MsgBox LBound(dizi)
MsgBox UBound(dizi)

End Sub

```

7. Diziyeye ait temel sayısal değerler aşağıdaki kodlar yardımı ile görüntülenebilir.

```

Sub diziler5()
Dim dizi(1 To 15) As Integer
Dim i As Integer

'Burada Excel hücrelerinde yer alan değerlerimizi
'dizi isimli dizimizin içerisine tanımlıyoruz.
For i = 1 To 15
dizi(i) = Cells(i, 1)
Next i

'dizinin temel değerleri aşağıdaki kodlar ile görüntülenebilir

MsgBox "Dizinin en büyük değeri " & Excel.WorksheetFunction.Max(dizi)
MsgBox "Dizinin en küçük değeri " & Excel.WorksheetFunction.Min(dizi)
MsgBox "Dizinin en büyük değeri " & Excel.WorksheetFunction.Average(dizi)
MsgBox "Dizinin en büyük değeri " & Excel.WorksheetFunction.Sum(dizi)

End Sub

```

8. Diziler aşağıdaki kodlarla istenilen karakterlere veya parçalara ayrılabilirler.

```

Sub diziler6()
Dim metin As String
metin = "VB10255VB100075VB68633"

Dim dizi As Variant
'metin değişkeni içerisindekileri "VB" karakterine ayır ve dizi içerisine a
1
dizi = Split(metin, "VB")

'Dizinin 1inci elemanını göster
MsgBox dizi(1)

```

```
End Sub
```

9. Diziler aşağıdaki kodlarda olduğu gibi filtrelenebilirler veya içerisindeki karakterlere göre aranıp getirilebilir.

```
Sub diziler7()
Dim isimler As Variant
Dim filtre As Variant

isimler = Array("mustafa", "mahmut", "mehmet", "ayşe", "mert", "oya")
filtre = Filter(isimler, "a", True, vbTextCompare)

End Sub
```

10. DİNAMİK DİZİLER

10.1. ELEMAN EKLEME

Excelde dizimizin son satırından sonra eleman ekledikçe dizimizin güncellenmesi.

```
Sub dinamikdizi()
Dim sayi_dizisi() As Integer
'son satırı buluyoruzki For döngüsünün son değerini verebilelim
sonsatir = Cells(Rows.Count, "A").End(xlUp).Row

'dizimizi eleman eklenebilir hale getiriyoruz
ReDim sayi_dizisi(1 To sonsatir)

'elemanları dizinin içerisine alıyoruz.
'eğer excelde sonsatırdan sonraki hücreye bir değer daha ekleyecek olursak
'dizimiz kendini bu sonsatıra göre güncellemiş olacak ve boyutu otomatik
'olarak değişecektir.
For i = 1 To sonsatir
'A sütununun i'nci satırını sayi_dizisi'nin i'nci elemanına ata
sayi_dizisi(i) = Range("A" & i)
Next i

End Sub
```

10.2. DİZİNİN ELEMAN SAYISINI ARTTIRMA-GENİŞLETME

```
Sub dizi_genislet()
Dim dizi() As Integer
'son satırı buluyoruzki For döngüsünün son değerini verebilelim
sonsatir = Cells(Rows.Count, "A").End(xlUp).Row

'dizimizi eleman eklenebilir hale getiriyoruz
ReDim dizi(1 To sonsatir)

'elemanları dizinin içerisine alıyoruz.
'eğer excelde sonsatırdan sonraki hücreye bir değer daha ekleyecek olursak
```

```

'dizimiz kendini bu sonsatıra göre güncellemiş olacak ve boyutu otomatik
'olarak değişecektir.
For i = 1 To sonsatir
dizi(i) = Cells(i, 1)
Next i

'dizinin elemanlarını topluyoruz
dizi_toplam = Excel.WorksheetFunction.Sum(dizi)

'dizideki elemanların toplamının 1000den küçük olması durumunda
'dizinin boyutunu 1 arttırarak son arttırılan hücreye bir değer daha
'girebiliriz. 1000 olduğunda da dizimiz duracaktır ve eleman eklememize
'izin vermeyecektir.
If dizi_toplam < 1000 Then
    ReDim Preserve dizi(1 To sonsatir + 1)
End If

End Sub

```

11. Ödev12: Exceldeki tabloyu 2 boyutlu dizi tanımlayıp içerisine atma.

	A	B	C	D	E	F
1	Test Adı	Değer				
2	ALT	850				
3	AST	500				
4	BUN	12				
5	Glikoz, Erişkin	115				
6	Kalsiyum	8				
7	Klor	110				
8	Kreatinin	5				
9	Kreatin Kinaz	300				
10	Kreatin Kinaz-MB	100				
11	Miyogloblin	250				
12	Protein	15				
13	Potasyom	4				
14	Sodyum	100				
15						
16						
17						
18	ALT	850				
19	AST	500				
20	BUN	12				
21	Glikoz, Erişkin	115				
22	Kalsiyum	8				
23	Klor	110				
24	Kreatinin	5				
25	Kreatin Kinaz	300				
26	Kreatin Kinaz-MB	100				
27	Miyogloblin	250				
28	Protein	15				
29	Potasyom	4				
30	Sodyum	100				

```

Sub dizi_ornek()
Dim dizi(14, 2) As Variant

```

```
Dim i As Integer
Dim j As Integer

'A2'den B14'e kalan tablo değerlerini diziye aktarıyoruz
For i = 2 To 14
    For j = 1 To 2
        dizi(i, j) = Cells(i, j).Value
    Next j
Next i

'kontrol için diziyi aşağı satırlarda bir yerde yazdırıyorum
For i = 1 To 14
    For j = 1 To 2
        Cells(i + 15, j) = dizi(i, j)
    Next j
Next i
End Sub
```

EXCEL İLE BAŞKA PROGRAMLARI KONTROL ETMEK

1.